Am I creating this LTSpice Symbol Correctly?
You are thinking in terms of creating a package-based symbol and that's leading to some confusion between the spice sub-circuit model you are also looking at.

Instead, the spice model you show is really laid out for a more normal opamp-like symbol for your schematic.

Instead, copy this file:

Version 4
SymbolType CELL
LINE Normal -32 32 32 64
LINE Normal -32 96 32 64
LINE Normal -32 32 -32 96
LINE Normal -28 48 -20 48
LINE Normal -28 80 -20 80
LINE Normal -24 84 -24 76
LINE Normal 0 32 0 48
LINE Normal 0 96 0 80
LINE Normal 4 44 12 44
LINE Normal 8 40 8 48
LINE Normal 4 84 12 84
WINDOW 0 16 32 Left 0
WINDOW 38 44 50 Left 0
SYMATTR SpiceModel LM358
SYMATTR Prefix X
SYMATTR Description National LM385
SYMATTR ModelFile LM358.MOD
PIN -32 80 NONE 0
PINATTR PinName In+
PINATTR SpiceOrder 1
PIN -32 48 NONE 0
PINATTR PinName In-
PINATTR SpiceOrder 2
PIN 0 32 NONE 0
PINATTR PinName V+
PINATTR SpiceOrder 3
PIN 0 96 NONE 0
PINATTR PinName V-
PINATTR SpiceOrder 4
PIN 32 64 NONE 0
PINATTR PinName OUT
PINATTR SpiceOrder 5


And save it as a file called LM358.ASY in some directory you can easily use. If you want, go into the LTspice installation directory, find the "lib" subdirectory there, then look inside that and find the "sym" subdirectory. Place the above file in there, if you can't place it anywhere else that is convenient.

However, I usually create my own symbol directory and place it there. This can then be accessed by first going to the Tools/Control Panel selection and then picking the "Sym. &amp; Lib. Search Paths" tab and adding your own personal directory in the Symbol Search Path box.

While you are here, you might as well also add a new directory of your own for the Spice simulation models in that box, as well. I keep my model files in one directory and my symbol files in a second one. But you could use one directory for both, too. Doesn't matter. LTspice just needs to be told about things, either way.

Then, restart the program after saving. Then when you hit F2, you can see at the very top that there is a "Top Directory" entry there and you will find your own personal directory listed there. Select that, then select LM358 from there.

Either way works.

However. As I mentioned above, you will ALSO need an LM358.MOD file and that needs to be placed into another directory (or the same directory.) If you look at the symbol definition included above, you will see a line that says:

SYMATTR ModelFile LM358.MOD


That's where the symbol points to the model file. So you have to use that name, or else change the symbol description above to use whatever name you want to use.

Then you need to store your model in such a file.

In my case, my model file says the following. But you can use your own model, as well. But I think you will be able to see where the nodes are also shown as just five nodes here, too. Just like yours.

The key to all this is that the node numbers match up with the symbol nodes. If you want, you can use LTspice to open up the LM358.ASY file, too. And then you can see the symbol there. Mouse over to a pin and right click on it. You will see the node number listed there and some other useful information. You can match these up (their numbers) with the order in which they are also found in the subcircuit node list in the model below here.

Go back and look up at the LM358.ASY file given above and note the "SpiceOrder" entries there. Five of them, right?

*//////////////////////////////////////////////////////////////////////
* (C) National Semiconductor, Inc.
* Models developed and under copyright by:
* National Semiconductor, Inc.  

*/////////////////////////////////////////////////////////////////////
* Legal Notice: This material is intended for free software support.
* The file may be copied, and distributed; however, reselling the 
*  material is illegal

*////////////////////////////////////////////////////////////////////
* For ordering or technical information on these models, contact:
* National Semiconductor's Customer Response Center
*                 7:00 A.M.--7:00 P.M.  U.S. Central Time
*                                (800) 272-9959
* For Applications support, contact the Internet address:
*  amps-apps@galaxy.nsc.com

*//////////////////////////////////////////////////////////
*LM358 DUAL OPERATIONAL AMPLIFIER MACRO-MODEL
*//////////////////////////////////////////////////////////
*
* connections:      non-inverting input
*                   |   inverting input
*                   |   |   positive power supply
*                   |   |   |   negative power supply
*                   |   |   |   |   output
*                   |   |   |   |   |
*                   |   |   |   |   |
.SUBCKT LM358       1   2  99  50  28
*
*Features:
*Eliminates need for dual supplies
*Large DC voltage gain =             100dB
*High bandwidth =                     1MHz
*Low input offset voltage =            2mV
*Wide supply range =       +-1.5V to +-16V
*
*NOTE: Model is for single device only and simulated
*      supply current is 1/2 of total device current.
*      Output crossover distortion with dual supplies
*      is not modeled.
*
****************INPUT STAGE**************
*
IOS 2 1 5N
*^Input offset current
R1 1 3 500K
R2 3 2 500K
I1 99 4 100U
R3 5 50 517
R4 6 50 517
Q1 5 2 4 QX
Q2 6 7 4 QX
*Fp2=1.2 MHz
C4 5 6 128.27P
*
***********COMMON MODE EFFECT***********
*
I2 99 50 75U
*^Quiescent supply current
EOS 7 1 POLY(1) 16 49 2E-3 1
*Input offset voltage.^
R8 99 49 60K
R9 49 50 60K
*
*********OUTPUT VOLTAGE LIMITING********
V2 99 8 1.63
D1 9 8 DX
D2 10 9 DX
V3 10 50 .635
*
**************SECOND STAGE**************
*
EH 99 98 99 49 1
G1 98 9 POLY(1) 5 6 0 9.8772E-4 0 .3459
*Fp1=7.86 Hz
R5 98 9 101.2433MEG
C3 98 9 200P
*
***************POLE STAGE***************
*
*Fp=2 MHz
G3 98 15 9 49 1E-6
R12 98 15 1MEG
C5 98 15 7.9577E-14
*
*********COMMON-MODE ZERO STAGE*********
*
*Fpcm=10 KHz
G4 98 16 3 49 5.6234E-8               
L2 98 17 15.9M
R13 17 16 1K
*
**************OUTPUT STAGE**************
*
F6 50 99 POLY(1) V6 300U 1
E1 99 23 99 15 1
R16 24 23 17.5
D5 26 24 DX
V6 26 22 .63V
R17 23 25 17.5
D6 25 27 DX
V7 22 27 .63V
V5 22 21 0.27V
D4 21 15 DX
V4 20 22 0.27V
D3 15 20 DX
L3 22 28 500P
RL3 22 28 100K
*
***************MODELS USED**************
*
.MODEL DX D(IS=1E-15)
.MODEL QX PNP(BF=1.111E3)
*
.ENDS
*$